

SAS/GRAPH gyakorlati bemutató SAS 9.1-hez

Tartalomjegyzék

1.	Környezet megismerése	2
1.1	Példa adatállomány megismerése	2
1.2.	SAS HELP	3
1.3	Eszközök grafikus riport készítésére	4
1.4	SAS/GRAPH programok általános szintaxisa	5
2.	Graph-N-Go	6
2.1	Oszlopdiagram készítése.....	6
2.2	Forráskód	8
2.3	Output	10
3.	Értelmezés.....	11
3.1	ODS.....	11
3.2	GOPTIONS.....	12
3.3	TITLE és FOOTNOTE	15
3.4	Tengely beállítások	16
3.5	Diagram beállítások	17
4.	Egyéb grafikonok.....	19
4.1	Pontdiagram	19
4.2	Vonaldiagram.....	22
4.3	Kördiagram	23
5.	Gyakorlatok.....	27
5.1	Gyakorlat 1.....	27
5.2	Gyakorlat 2.....	27
5.3	Gyakorlat 3.....	28

László Anna
2010. április 26.


1. Környezet megismerése

1.1 Példa adatállomány megismerése

Vegyük példának a **Sashelp** könyvtár **Class** nevű adatállományát a SAS/GRAPH megismeréséhez. Másoljuk először a Sashelp.Class adatállományt a saját (graph) könyvtárunkba pl. c névvel:

```
data graph.c ;
  set sashelp.class ;
run ;
```

Ismerjük meg az adatállományunk!



	Name	Sex	Age	Height	Weight
1	Alfred	M	14	69	112.5
2	Alice	F	13	56.5	84
3	Barbara	F	13	65.3	98
4	Carol	F	14	62.8	102.5
5	Henry	M	14	63.5	102.5
6	James	M	12	57.3	83
7	Jane	F	12	59.8	84.5
8	Janet	F	15	62.5	112.5
9	Jeffrey	M	13	62.5	84
10	John	M	12	59	99.5
11	Joyce	F	11	51.3	50.5
12	Judy	F	14	64.3	90
13	Louise	F	12	56.3	77
14	Mary	F	15	66.5	112
15	Philip	M	16	72	150
16	Robert	M	12	64.8	128
17	Ronald	M	15	67	133
18	Thomas	M	11	57.5	85
19	William	M	15	66.5	112

Ha megnyitjuk a graph.c adatállományt, láthatjuk, hogy 5 változónk és 19 megfigyelésünk/egyedünk van. További információkat tudhatunk meg a contents eljárás futtatásával:

```
proc contents data=graph.c ;
run ;
```

A változók jellemzőit pl. az alábbi táblázat tartalmazza az eredményben (HTML outputban: Tools → Options → Preferences → Results → Create HTML):

Változók és attribútumok alfabetikus listája			
#	Változó	Típus	Hossz
3	Age	Num.	8


Változók és attribútumok alfabetikus listája			
#	Változó	Típus	Hossz
4	Height	Num.	8
1	Name	Kar.	8
2	Sex	Kar.	1
5	Weight	Num.	8

Ebből látható, hogy a változók nem lettek címkével (label) ellátva, vagyis, ami változó elnevezéseket láttunk előzőleg az adattábla megnyitásakor, ugyanazok maguk a változónevek.

Az 5 változóból **3 nomenklátúra** (Age, Name, Sex) és **2 mutató** (Height, Wight). (Megjegyzés: Az Age változó értelmezhető mutatóként is.)

1.2. SAS HELP

A 9.1-es SAS-ban a SAS/GRAPH eljárások listáját a Help → SAS Help and Documentation → Tartalom → SAS Products → SAS/GRAPH → SAS/GRAPH Reference → SAS/GRAP H Procedures részben találjuk:



1.3 Eszközök grafikus riport készítésére

- SAS/GRAPH
- SAS Enterprise Guide Graphical Tasks and Wizards
- SAS/AF (Alkalmazásfejlesztés, jelentősége EM-mel és Java-val ma már háttérbe szorult)
- JMP (Statisztikai programcsomag)
- SAS/INSIGHT, SAS Forecast Studio, SAS Enterprise Miner (EM)
- SAS Platform for Business Analytics (Web OLAP Viewer)
- Graph-N-Go

1.4 SAS/GRAPH programok általános szintaxisa

```
OPTIONS options; /* SAS opciók beállítása */
GOPTIONS options; /* grafikus opciók beállítása */
<DATA step (ha szükséges);> /* lehet, hogy szükséges előbb módosításokat végezni az
adatállományon, vagy pl. rendezni: PROC SORT */
<SAS/GRAPH global statements;> /* TITLE, FOOTNOTE, AXIS, LEGEND, stb. */
PROC <SAS/GRAPH procedure name> DATA=libref.dataset;
  Procedure action statements;
RUN;
QUIT;
```

2. Graph-N-Go

2.1 Oszlopdiaagram készítése

Indítsuk el a Graph-N-Go-t: Solutions → Reporting → Graph-N-Go


A bal oldalon lévő ikonsorból húzzuk a „New SAS Data Set” ikont a felső ablakrészbe.

A „SAS Data Set” sorban válasszuk a graph.c adatállományunkat (a különböző munkalapokon ellenőrizhetjük az állományunk jellemzőit).


Húzzunk az egerrel pl. egy „New Bar Chart” diagramot az alsó ablakrészbe. Az üres diagramterületen dupla egérekattintással beállíthatjuk a kívánt oszlopdiaagram kinézetünket.

Legyen pl. az alábbi:


- „Data” fül alatt válasszuk a graph.c adatállományt, és tegyük meg a következő beállításokat:
 - Kor legyen a kategória változó
 - Fügő változónak állítsuk a *Tömeget*
 - Csoportképző ismérvünk legyen a *Nem*
 - Az *átlagot* szemléltessük statisztikai értéként



- „Titles/Footnotes” fül alatt:
 - Cím legyen mondjuk „*BAR Chart of Weight*”
 - Lábjegyzet: „*Age grouped by Sex*”




- „Appearance” fül alatt:
 - Color Scheme pl. *International*
 - Használjunk az ábrán *referencia vonalakat*
 - Az oszlop típusa legyen pl. *Cylinder*




- „Object Size” fül alatt állítsuk be a diagramterület méretét 250×300-ra! (A méretet a későbbiekben úgy is módosíthatjuk, hogy a diagram területén jobb egérgomb lenyomásával megjelenő

menüsorban kiválasztjuk a „Grow/Shrink” opciót, és manuálisan beállítjuk a kívánt méretet. A „Maximize” menüponttal teljes nagyítást is végezhetünk.)



- A többi beállítás maradjon az alapértelmezett értéken.
Az OK gomb lenyomásával az alábbi grafikont kapjuk:



2.2 Forráskód

Az Export → „Source file” menüpont segítségével a kapott diagramot generáló SAS kódot kimenthetjük. A megjelenő ablakban „Preview” gomb lenyomásával megjelenik egy külön ablakban a forráskód. Másoljuk be az Editor ablakba, és futtassuk le:

```
/* Graph-N-Go SAS/Graph Code Generation for
**
** Viewer:      Bar Chart 1 18APR2010 12:25:46
** Description: Bar Chart 1
** Modified:    18APR2010:14:49:57
**
```



```

** SAS products required: Base, SAS/Graph (Version 9 or later)
** Code generated on: 18APR2010 14:56:21
**
** There may be differences in appearance of the graph
** generated by the code below and the Graph-N-Go viewer.
**
** To make code modifications consult the documentation
** for these statements: ODS, GCHART, GPLOT,
** GOPTIONS, AXIS, LEGEND, SYMBOL, TITLE, FOOTNOTE.
**
** To route output to a graphics device other than your monitor,
** modify the source code below to change the device driver by
** 1) removing the asterisk preceding GOPTIONS DEVICE=JAVA;
** 2) changing JAVA to some other valid device.
**
** To create an HTML file, modify the source code below
** to enable ODS output by
** 1) removing the asterisks from the ODS statements and
** the asterisk preceding GOPTIONS DEVICE=JAVA;
** 2) setting the value of DEVICE= to GIF, JAVAIMG, ACTXIMG,
** JAVA, or ACTIVEX (JAVA and ACTIVEX create interactive graphs)
** 3) verifying or changing the ODS FILE= option so it names the
** HTML file for output.
** 4) verifying or changing the ODS GPATH= option so it names the
** path for the GIF or JPG file displayed by the HTML page. This
** option is not necessary if the DEVICE is ACTIVEX or JAVA.
*/

/* Begin ODS output */
* ods listing close;
* ods html file="output-HTML-file-specification" (title="Graph-N-Go Output")
      gpath="output-image-path-specification" gtitle gfootnote style=minimal;

/* Set the SAS/Graph options */
options reset=all
      ctext=CX800080 ftext="MS Sans Serif" htext=8 pt
      colors=(CX13967A CX4175B0 CXFFE4AB CXFF7370
              CX204776 CXDD9B2B CXE64E4B CXA5A5A5
              CX754B25 CXD63A3C CX688CD6 CXFFFFFF
              CX444444 CXB92C2C CX008080 CXFDDD53);

/* Set the Titles/Footnotes */
title1 justify=center color=CX000000 font="MS Sans Serif" height=8 pt "BAR Chart of
Weight";
footnote1 justify=center color=CX000000 font="MS Sans Serif" height=8 pt "Age grouped
by Sex";

/* Set the SAS/Graph device driver */
* options device=JAVA xpixels=444 ypixels=335;

/* AXIS1 describes axis for Category variable AGE */
/* AXIS2 describes axis for Response variable WEIGHT and statistic MEAN */
axis1 minor=none noplane label=("Age");
axis2 minor=none noplane label=("Weight (Average)")
      order=(0 to 150 by 50) ;

proc gchart data=GRAPH.C;
  vbar3d AGE /
    type=MEAN sumvar=WEIGHT shape=CYLINDER
    maxis=axis1 noframe autoref clipref cautohref=CXC0C0C0
    woutline=1 coutline=CX000000 caxis=CX000000

```

```


axis=axis2 group=SEX G100 ;
run;
quit;

/* Reset all graphics options */
options reset=all;

/* End ODS output */
* ods html close;
* ods listing;

```

2.3 Output



Az outputokat a WORK.GSEG katalógusba menti a SAS.

3. Értelmezés

3.1 ODS

ODS = Output Delivery System

SAS kimenet.

A SAS táblázatos vagy grafikus kimenetét többféle kimenetre irányíthatjuk.

Alapértelmezetten LISTING (SAS GRAPH1 Window), de lehet HTML (Web Browser), PDF (Adobe Reader, Printer, ...), RTF (Word Processor), stb. kimeneteket is beállítani.

Szintaxis:

```
ods listing;
...
ods listing close;
```

Minden korábban beállított kimenetet egy lépésben is lezárhatunk:

```
ods _all_ close;
```

Példa:

```
ods listing;
ods html file='d:\SAS\2010_I\gyak\SAS_GRAPH\ods\c.html';
ods rtf file='d:\SAS\2010_I\gyak\SAS_GRAPH\ods\c.rtf';
ods pdf file='d:\SAS\2010_I\gyak\SAS_GRAPH\ods\c.pdf';
ods noptitle;


goptions reset=all;

proc freq data=graph.c;
  title 'Age frequencies in the Class';
  tables age / nocum nopercent;
run;

proc gchart data=graph.c;
  vbar age;
  where sex='F';
  title 'Age frequencies among girls';
run;

quit;
ods _all_ close;
title;
```

Megjegyzés: a html file-ban nem minden esetben jeleníti meg az ábrát, mert azt külön képként menti le. Ha a „file=” opciót kihagyjuk, akkor kirajzolja a grafikont is a böngészőben. A lényeg, hogy, ha megnyitjuk valamelyik ods-t, akkor zárjuk is le a kódsor végén, különben nem jeleníti meg az ábrát!



Töröljük ki a program elejéről és végéről a kikommentezett (*) részeket a kimeneti beállításokra vonatkozóan. Hogy web böngészőben is megnézhessek a grafikonunkat, állítsuk be a Tools → Options → Preferences → Results → Preferred web browser opciót.

Futtassuk újra a programot és web böngészőben nézzük meg a kimenetet. (Megkereshetjük, hogy hova menti le a SAS a képet a számítógépünkre.)

3.2 GOPTIONS

Globális grafikus alapbeállításokat lehet ezzel tenni.

A QUIT; utasítás zárja le.

Több eljárás is lehet benne.

```
GOPTIONS graphics-specific-options; /* több opció is lehet itt */
PROC graphics-procedure-name DATA=libref.dataset;
    Statements specific to the graphics procedure
    / <statement options>;
RUN;
QUIT;
```

A GOPTIONS-ben megjelenő néhány opció vonatkozhat általános beállításra vagy visszaállításra (RESET=ALL | GLOBAL, ...), a megjelenésre (BORDER | NOBORDER, GSIZE=lines, HPOS=columns, IBACK=fileref | 'external-file', VPOS=rows, ...), színre (CBACK=background-color, CTEXT=text-color, CTITLE=title-color, COLORS=<(colors-list | NONE)>, ...), szöveg kinézetére (FTEXT=text-font, FTITLE=title-font, HTEXT=text-height <units>, HTITLE=title-height <units>, ...), stb. (További leírás a „The GOPTIONS Procedure” Help-jében.)


A színekre vonatkozóan az alábbi oldalon is találunk leírást:
<http://www.tau.ac.il/cc/pages/docs/sas8/gref/zgscheme.htm>

```
goptions reset=all ftext='Arial/bold';  
ods listing;
```


```
proc sort data=graph.c out=csex;  
  by sex;  
  where age in (11,12,13);  
run;
```

```
proc gchart data=csex;  
  by sex;  
  vbar height;  
  title 'Heights of 11-13 year old boys and girls';  
run;  
quit;
```

Heights of 11-13 year old boys and girls



Heights of 11-13 year old boys and girls



3.3 TITLE és FOOTNOTE

Szintaxis

```
TITLEn options 'text';
FOOTNOTEn options 'text';
```

„n” 1 és 10 közötti egész szám lehet, ha kihagyjuk, alapértelmezetten az első címet vagy lábjegyzetet jelenti.

Opciók

FONT= or F=

```
title f='Albany AMT' 'Cím1';
```

COLOR= or C=

```
title c=blue 'Cím1';
```

HEIGHT= or H=

```
title h=14pt 'Cím1';
```

JUSTIFY= or J=

```
title j=center 'Cím1';
```

UNDERLINE= or U=

```
title u=2 'Cím1';
```


BOX= or BO=

```
title box=2 'Cím1';
```

Példa:

```
options reset=all;
ods listing;

proc gchart data=GRAPH.C;
  HBAR AGE;
  TITLE f='Times New Roman' c=blue h=14pt j=center u=2 box=2 'Cim1';
  TITLE2 f='Times New Roman' c=green h=12pt j=center u=1 'Cim2';
  Footnote f='Times New Roman' c=cyan h=10pt j=center u=3 box=1 'Labjegyzet1';
run;
quit;
```



Csak a cím és lábjegyzet grafikus beállítások törlése:

```
options reset=(title footnote);
```

Minden grafikus beállítás törlése:

```
options reset=global;
```

3.4 Tengely beállítások

Szintaxis:

```
AXISn options;
```

Opciók:

```
ORDER=(value list) /* értékek sorrendje a tengelyen */
```

```
OFFSET=(n1 units, n2 units) | (n1,n2)units
```

```
axis1 order=(1 to 6) offset=(1 in, 1 cm); /* az origótól 1 inch-re, a végponttól 1 cm-re, egység lehet: CELLS (character cells), CM (centiméter), IN (inch), PCT (grafikus kimeneti terület százaléka), PT (pont) */
```

```
axis2 order=(120 to 320 by 20);
```

```
axis3 order=(100 200 300);
```

```
axis4 order=('Spain' 'Italy' 'France') offset=(3,3)cm;
```

```
LENGTH=n /* tengely hossza */
```

```
ORIGIN=(x units, y units) | (x,y)units /* a tengely hol lássa az origót, origó elhelyezkedésének beállítása*/
```

```
axis5 length=60pct origin=(20,15)pct;
```

```
axis6 length=6in origin=(1in,20pct);
```

```
MAJOR=(tick-mark-options) | NONE /* főosztások, lehetséges opciók: NUMBER=, COLOR=, HEIGHT=, WIDTH= */
```



```
MINOR=( tick-mark-options) | NONE /* alosztások, lehetséges opciók: NUMBER=, COLOR=,
HEIGHT=, WIDTH= */
axis1 major=(number=12) minor=none;
axis2      major=(color=red height=2 width=2)
           minor=(color=orange height=1 number=1)
           order=(120 to 320 by 20);
```

```
LABEL=(text-options 'text') | NONE /* tengely címke testre szabása, lehetséges opciók: COLOR=,
HEIGHT=, FONT= */
```

```
VALUE=(text-options-n 'text-n') | NONE /* lehetséges opciók: COLOR=, HEIGHT=, FONT=, TICK= */
axis1 label=(color=blue height=2 font='Courier New' 'Budget Month')
           value=(color=red height=1.5 font='Arial');
axis2 label=none;
axis3 value=(color=orange
           t=1 'JAN' t=2 color=red 'FEB'
           t=3 'MAR' t=4 color=red 'APR'
           t=5 'MAY' t=5 color=red 'JUN'); /* TICK=t */
```

GPLOT Procedure

HAXIS: Horizontal Axis – vízszintes tengely

VAXIS: Vertical Axis – Függőleges tengely

Példa:

```
ods html;

goptions reset=global;
symbol v=square cv=green h=10pt; /* négyzet formájú szimbólum a grafikonon */

axis1 order=(10 12 14 16 18)
       offset=(1,1)cm length=80pct origin=(, 14.3 pct)
       major=(height=1 width=2)
       minor=none
       label=(height=2pct font='Arial' color=green 'Age for Boys');
axis2 major=(height=2 width=2)
       minor=(number=4 color=red height=1)
       label=(height=2pct font='Arial' color=green)
       value=(font='Courier New');

proc gplot data=graph.c;
  where sex='M' and weight>90;
  plot height*age / haxis=axis1 vaxis=axis2
                  noframe autohref
                  vref=65; /* referencia vonal */
  format height comma6.;
  title 'Height of boys with Weight greater than 90 by Age';
run;
quit;

ods html close;
```

3.5 Diagram beállítások

(SAS Help: BAR statement, GBARLINE procedure)

```
vbar3d AGE /
  type=MEAN sumvar=WEIGHT
  shape=CYLINDER
  maxis=axis1 caxis=CX000000
  raxis=axis2
  noframe
  autoref clipref cautohref=CXC0C0C0
```

```
woutline=1 coutline=CX000000  
group=SEX G100;
```

type: alkalmazott statisztika típusa

sumvar: változó kiválasztása, amire a statisztikát számoljuk

shape: oszlopok formájának típusa az oszlopdiagramon. Lehet pl. block=B, cylinder=C, prism=P, star=S, hexagon=H.

maxis: vízszintes főtengely kijelölése

raxis: „response axis”, bal oldali függőleges tengely

caxis: tengely színe

noframe: ne rajzoljon keretet a diagram köré

autoref: a függőleges tengely minden főosztásához rajzol egy-egy referencia egyenest. A függőleges tengely adott pontjához való referencia egyenes rajzolásához használjuk a REF= opciót.

clipref: alapértelmezetten a referencia vonalak az oszlopok előtt futnak. Hogy az oszlopok mögött fussanak a referencia vonalak, használjuk a CLIPREF opciót.

cautoref: referencia egyenes színe

woutline: oszlop kontúr vastagsága pixelben (pl. block típusú oszlopformánál látszódik)

coutline: oszlop kontúrvonalának színe

group: csoportosító változó kijelölése

g100: a csoportváltozó szerinti tengelyosztások egyenletes eloszlását, és a tengely 100 százalékának kihasználtságát ezzel adjuk meg.

4. Egyéb grafikonok

4.1 Pontdiagramm


Scatter Plot, Kavicsábra

Szintaxis:

```
proc gplot data=SAS-data-set; /* bemeneti adathalmaz */
  plot y-variable * x-variable /* előbb a függő változó, majd a független */
  / options;
run;
quit;
```

Példa:

```
proc gplot data=graph.c;
  plot weight * height;
run;
quit;
```



Alapértelmezetten a „pontok” szimbóluma plusz-jel.

A szimbólumokat az alábbi módon lehet módosítani:

```
Symboln value=symbol cv=color height=n units;
/* value=v=szimbólum, cv=szimbólum színe,
height=h=szimbólum magassága alapértelmezetten CELL, de lehet CM, IN, PT, PCT */
```


Néhány szimbólum: plus, Y, star, square, diamond, triangle, hash, Y, Z, paw, point, dot, circle, _, \$, +, :,


...

(SAS HELP: SYMBOL statement, options)

Példa:

```
symbol1 value=diamond cv=magenta height=25 pt;  
symbol2 value=dot cv=green height=0.2 in;  
proc gplot data=graph.c;  
  plot height * weight=1 /* symbol1 beállítással */  
      height * age=2; /* symbol2 beállítással */  
run;  
quit;
```






Példa néhány további opció alkalmazására:

```

symbol1 value=diamond cv=magenta height=25 pt;
*symbol2 value=dot cv=green height=0.2 in;
proc gplot data=graph.c;
  plot height * weight=1 /* symbol1 beállítással */
        /*height * age=2 /* symbol2 beállítással */
        / cframe=yellow
          grid;
  title 'Scatter plot with background color and grid';
run;
quit;

```

Scatter plot with background color and grid



4.2 Vonaldiagram

Line plot

A szintaxis alapja a pontdiagram szintaxisa, melyben a SYMBOL utasításban az INTERPOL= opciót kell használni a pontok „összekötésének” stílusára.

INTERPOL=

NONE: nincs vonal, ami összeköti a pontokat, ez az alapértelmezett

JOIN: egyenes szakaszokkal köti össze a pontokat a pontdiagramon

SPLINE: simított, lekerekített vonallal köti össze a pontokat

SMnn: simított vonal, ami nem feltétlen megy át a pontokon. „nn” 0 és 99 közötti egész szám, minél nagyobb, annál jobban simul a pontokhoz illesztett görbe.

Egyéb opciók:

WIDTH=n (w=n)

vonaltvastagság beállítása

LINE=line-type (l=line-type)

vonaltípusa (pl. szaggatott, pontozott, stb.), 1-46 közötti egész szám

CI=line-color

vonalszíne

COLOR=color (c=color)


a vonal és a pontok (szimbólumok) színének beállítása

Vonal típusait megtaláljuk a „*LINE= option, SYMBOL statement*”-re keresve a HELP-ben.

Példa:

```
symbol1 value=square cv=blue height=0.5 cm
      interpol=join width=2 line=10 ci=green;
proc gplot data=graph.c;
  plot height * name =1
    / grid;
  title 'Line plot for heights';
run;
quit;
```

Line plot for heights



4.3 Kördiagram

Pie chart, tortadiagram

Szintaxis:


```
proc gchart data=SAS-data-set;
  pie chart-variable(s) / options;
  pie3d chart-variable(s) / options;
run;
quit;
```

Példa:

```
proc gchart data=graph.c;
  pie weight;
  pie3d weight;
  title 'Pie charts';
run;
quit;
```


Pie charts

FREQUENCY of Weight



Pie charts

FREQUENCY of Weight



Néhány lehetséges opció:

TYPE=statistic	A számítandó statisztika, pl. FREQ (alapértelmezett), PERCENT, SUM
SUMVAR=variable	Az összeg számításához használt változó megnevezése
GROUP=variable	Csoportosító változó; annyi kördiagram lesz, ahány csoport van
SUBGROUP=variable	Alcsoportot meghatározó változó, amely alcsoportok a kördiagramon koncentrikus körökként jelennek meg
LEVELS=n	Csoportok számának meghatározása

A tortaszeletek megjelenése:

NOHEADING	Alapértelmezetten „3 óránál” kezdi a körcikkelyeket rajzolni óramutató járásával ellentétes irányba
ASCENDING	A statisztikai érték szerint növekvő sorrendbe teszi a cikkelyeket
DESCENDING	A statisztikai érték szerint csökkenő sorrendbe teszi a cikkelyeket
CLOCKWISE	„12 óránál” kezdi a körcikkelyeket rajzolni az óramutató járásával megegyező irányba
ANGLE=value	A megadott értéknél, mint foknál kezdi rajzolni az első tortaszeletet óramutató járásával ellentétes irányba. A 0 fok „3 óránál” van.

Értékek megjelenítése:


ARROW	Kívül a körön, nyíllal kötve a cikkelyhez
INSIDE	Körcikkelyen belül
OUTSIDE	Körcikkelyen kívül, nem összekötve a tortaszelettel, csak mellette elhelyezve
NONE	Nem jeleníti meg az értékeket.

PERCENT	Cikkelyhez tartozó százaléértékek megjelenítése
SLICE	Cikkely nevének (középső értékének) megjelenítése
VALUE	Cikkelyhez tartozó statisztikai érték megjelenítése


Példa:

```
proc gchart data=graph.c;
  pie3d age
    /   type=freq
        group=sex
        clockwise
        value=outside;
run;
quit;
```

FREQUENCY of Age



FREQUENCY of Age



5. Gyakorlatok

5.1 Gyakorlat 1

Feladat:


Rajzoljunk egy pontdiagramot 20 pont méretű csillag szimbólumokkal ahol szemléltetjük az osztályban (sashelp.class) a 15 évnél fiatalabbak között magasság értékeket! Adjunk címet a diagramnak!

Megoldás:

```
options reset=all;
symbol1 value=star cv=blue height=20 pt;

proc gplot data=graph.c;
  where age<15;
  plot height * age =1;
  title 'Scatter plot for heights of students under 15 years of age';
run;
quit;
```

Scatter plot for heights of students under 15 years of age



5.2 Gyakorlat 2

Feladat:

Készítsünk olyan simított vonaldiagramot, amiben az osztály (sashelp.class) fiú tagjainak tömeg értékeit szemléltetjük csökkenő sorrendbe állítva a diákokat! A vonal legyen szaggatott és megegyező színű a pontokkal! Állítsunk be valamilyen háttérszínt és adjunk címet a grafikonunknak!

Megoldás:

```
proc sort data=graph.c out=cc;
  where sex='M';
  by descending weight;
run;
```


```

data cc;
  set cc;
  do i=1 to _n_-1;
    i=_n_-1;
  end;
  label i='boys';
run;

symbol1 value=+ height=0.8 cm
          interpol=spline width=2 line=3 color=yellow;
proc gplot data=cc;
  plot weight * i =1
    / cframe=blue;
  title 'Smoothed line graph for boys' weights in descending order';
run;
quit;

```

Smoothed line graph for boys' weights in descending order



5.3 Gyakorlat 3

Feladat:

Hozzunk létre egy vízszintes oszlopdiagramot a sashelp.class adatállományon, amiben a magasságok gyakoriságát mutatjuk be nemek szerint! Használjunk fekete címeket és piros lábjegyzeteket középre rendezve! Az ábra szövegezését kék színnel, Arial betűtípusban, 10 pont betűmérettel jelenítsük meg! A tengelyek színe legyen zöld, az oszlopok kontúrvonala pedig legyen 4 pont vastag és kék színű! A grafikont pdf kimenetre irányítsuk! Ne felejtsük el a végén visszaállítani a kezdeti beállításokat!

Megoldás:

```

ods pdf file='d:\SAS\2010_I\gyak\SAS_GRAPH\ods\c2.pdf';
options reset=all
          ctext=blue ftext="Arial" htext=10 pt;

```

```

title1 justify=center color=black font='Times New Roman' height=20 pt 'Horizontal bar
chart of Height by gender';
title3 justify=center color=black font='Times New Roman' height=15 pt
'SASHELP.CLASS';
footnote1 justify=center color=red height=8 pt '2010';
footnote3 justify=center color=red height=8 pt 'Szeged';

proc gchart data=graph.c;
  hbar height
    / type=mean
      caxis=green
      group=sex
      woutline=4 coutline=blue;

run;
quit;

options reset=all;
ods pdf close;

```

